

Curve polari

di Attilio Scifoni - Roma

Una curva può essere rappresentata in forma cartesiana con un'equazione del tipo:

$$y = f(x)$$

ove x e y sono le coordinate cartesiane, rispettivamente ascissa e ordinata che individuano un punto della curva nel modo indicato in figura 1.

Oltre a questa forma esiste anche un'altra forma, detta forma polare, in cui una curva è rappresentata da un'equazione del tipo:

$$R = f(F)$$

ove R e F sono le coordinate polari, rispettivamente raggio vettore e anomalia.

Anch'esse individuano un punto nel modo indicato nella figura 2.

Ad esempio se l'equazione è:

$$R = a \cdot \sin^2 F / \cos F$$

avremo la cosiddetta Cissoide di Diocle che, prendendo per il parametro a il valore 200, è rappresentata in figura 3.

Analogamente se l'equazione è:

$$R = a \sqrt{\cos 2F + b}$$

avremo la cosiddetta Lemniscata di Bernoulli che, prendendo per il parametro a il valore 70 e per il parametro b il valore 0,5, è rappresentata in figura 4.

Programma Polare1

Per tracciare una curva a partire dalla sua forma polare basta, per F che varia da 0° a 360° , calcolare il corrispondente

È disponibile, presso la redazione, il disco con il programma presentato in questa rubrica. Le istruzioni per l'acquisto e l'elenco degli altri programmi disponibili sono a pag. 263.

R e tracciare il punto di coordinate polari (R, F) .

Ciò si ottiene col semplice programma illustrato nel listato 1. Digitata nella linea 30 l'equazione prescelta e introdotti, dopo il suo lancio, i valori prescelti dei parametri, esso traccia la curva polare corrispondente.

Se consideriamo la curva di equazione

$$R = a \cdot \cos F + b$$

(Lumaca di Pascal) dobbiamo digitare la linea

$$30 R = A \cdot \cos(F) + B$$

e, lanciato il programma, introdurre i valori prescelti per a e b . Ad esempio per $a = 40$ e $b = 40$ abbiamo la curva di figura 5, per $a = 30$ e $b = 40$ quella di figura 6 e per $a = 60$ e $b = 30$ quella di figura 7.

Se consideriamo la curva di equazione

$$R = a \cdot \sin \cos \sin b \cdot F$$

(Curva Lobata) dobbiamo digitare la linea

$$30 R = A \cdot \sin(\cos(\sin(B \cdot F)))$$

e, lanciato il programma, introdurre i valori prescelti per a e b . Ad esempio per $a = 80$ e $b = 2$ abbiamo la curva di figura 8, per $a = 80$ e $b = 3$ quella di figura 9 e per $a = 80$ e $b = 4$ quella di figura 10.

Programma Polare2

Possiamo modificare il programma precedente predisponendo il tracciamento delle curve con una linea continua invece che per punti. Inoltre possiamo aggiungere un controllo di errore che eviti il blocco dell'elaborazione quando i risultati dei calcoli sono molto grandi o non reali.

Il programma così modificato è indicato nel listato 2.


Se consideriamo la curva

$$R = a / \cos F + b$$

(Concoide di Nicomede) dobbiamo digitare la linea

$$30 R = A / \cos(F) + B$$





e, lanciato il programma, introdurre i valo-



ri prescelti per a e b. Ad esempio per a=50 e b=35 abbiamo la curva di figura 11, per a=50 e b=50 quella di figura 12 e per a=40 e b=70 quella di figura 13.

Se consideriamo la curva

$$R = a \cdot \cos b \cdot F$$



```

10 GRAPHIC0,1:INPUT"A,B":A,B:GRAPHIC1,1
20 FORG=OTO360:F=G*π/180
30 R=A*COS(F)+B
40 LOCATE160,100:DRAW,ABS(R):G+90-180*(R<0)
50 NEXTG

```

Listato 1

(Curva Rodonea) dobbiamo digitare la linea

$$30 R = A \cdot \cos(B \cdot F)$$

e, lanciato il programma, introdurre i valori prescelti per a e b. Ad esempio per a = 80 e b = 2 abbiamo la curva di figura 14, per a = 80 e b = 3 quella di figura 15 e per a = 80 e b = 4 quella di figura 16. Se consideriamo la curva

$$R = a/\cos b \cdot F$$

(Curva Spigata) dobbiamo digitare la linea

$$30 R = A/\cos(B \cdot F)$$

e, lanciato il programma, introdurre i valori prescelti per a e b.

Ad esempio per a = 40 e b = 2 abbiamo la curva di figura 17, per a = 30 e b = 3 quella di figura 18 e per a = 25 e b = 4 quella di figura 19.

Possiamo accelerare il tracciamento della curva ponendo un passo maggiore di uno nella linea 20, ad esempio ponendo in essa FORG = OTO360STEP5, a scapito però dell'accuratezza del tracciamento, dato che al crescere del pas-

```

5 TRAP100
10 GRAPHIC0,1:INPUT"A,B":A,B:GRAPHIC1,1
20 FORG=OTO360:F=G*π/180
30 R=A/COS(F)+B
40 LOCATE160,100:DRAW,ABS(R):G+90-180*(R<0)
42 Z=RDOT(0):W=RDOT(1)
44 IFF>0THENDRAW,X,YTOZ,W
46 X=Z:Y=W
50 NEXTG
100 RESUME46

```

Listato 2

```

5 TRAP100
10 GRAPHIC0,1:INPUT"A,B":A,B:GRAPHIC1,1
20 FORG=OTO360:F=G*π/180
30 R=A*COS(B*COS(F))
40 LOCATE160,100:DRAW,ABS(R):G+90-180*(R<0)
42 Z=RDOT(0):W=RDOT(1)
44 IFF>0THENDRAW,X,YTOZ,W
46 X=Z:Y=W
50 NEXTG
60 B=B+π/8:SLEEP2:SCNCLR:GOTO20
100 RESUME46

```

Listato 3

```

1 FORA=4864TO5001:READA$:POKEA,DEC(A$):NEXT
2 DATA EA,EA,E6,FC,D0,02,E6,FD,A9,00,85,FA,A9,20,85,FB,A0,00,EA,B1,FA,C9,00,F0,2
0,EA,A5,FA,91,FC,E6,FC,D0,02,E6,FD,A5,FB,91,FC,E6,FC,D0,02,E6,FD,B1,FA,91,FC
3 DATA E6,FC,D0,02,E6,FD,EA,E6,FA,D0,D6,E6,FB,A5,FB,C9,40,D0,CE,E6,FC,D0,02,E6,F
D,E6,FC,D0,02,E6,FD,A9,00,91,FC,60,00,EA,EA,E6,FA,D0,02,E6,FB,A0,00,EA,B1,FA
4 DATA 85,FC,E6,FA,D0,02,E6,FB,B1,FA,85,FD,EA,E6,FA,D0,02,E6,FB,B1,FA,D0,01,60,9
1,FC,EA,E6,FA,D0,02,E6,FB,A2,00,F0,D9,00
5 TRAP100
10 GRAPHIC0,1:INPUT"A,B":A,B:GRAPHIC1,1
15 POKE252,0:POKE253,69
20 FORG=OTO360:F=G*π/180
30 R=A*COS(B*COS(F))
40 LOCATE160,100:DRAW,ABS(R):G+90-180*(R<0)
42 Z=RDOT(0):W=RDOT(1)
44 IFF>0THENDRAW,X,YTOZ,W
46 X=Z:Y=W
50 NEXTG
55 BANK0:SYS4864
60 B=B+π/8:K=K+1:SCNCLR:IFK<20GOTO20
70 POKE250,0:POKE251,69:GRAPHIC1,1
80 FORK=1TO20:SCNCLR:BANK0:SYS4951:FORH=OTO100:NEXTH,K
90 GOTO70
100 RESUME46

```

Listato 4

so la curva tende a divenire una spezza-
ta di segmenti.

Programma Polare3

Possiamo modificare il programma precedente considerando il tracciamento di una famiglia di curve, cioè di un insieme di curve ottenute facendo variare un parametro.

Per ottenere ciò basta aggiungere al programma Polare2 la seguente linea:

$$60 B + B = \pi/8:SLEEP2:SCNCLR:GOTO20$$

Il programma così modificato è indicato nel listato 3. Se consideriamo le curve

$$R = a \cdot \cos(b \cdot \cos F)$$

(Curve Butterfly) dobbiamo digitare la linea

$$30 R = A \cdot \cos(B \cdot \cos(F))$$

e, lanciato il programma, introdurre il valore prescelto per a e il valore iniziale prescelto per b.

Ad esempio introducendo a = 80 e b = 1.18 il programma tratterà la famiglia di curve per il dato a e per b = 1.18, b = 1.57, b = 1.96 e così via.

Nelle figure 20, 21 e 22 sono indicate le prime tre curve di questa famiglia.

Se consideriamo le curve

$$R = a \cdot \sin(b \cdot \sin F)$$

(Curve Mosquito) dobbiamo digitare la linea

$$30 R = A \cdot \sin(B \cdot \sin(F))$$

e, lanciato il programma, introdurre il valore prescelto per a e il valore iniziale prescelto per b.

```

1 FORA=4864TO5001:READA$:POKEA,DEC(A$):NEXT
2 DATA EA,EA,E6,FC,D0,02,E6,FD,A9,00,85,FA,A9,20,85,FB,A0,00,EA,B1,FA,C9,00,F0,2
0,EA,A5,FA,91,FC,E6,FC,D0,02,E6,FD,A5,FB,91,FC,E6,FC,D0,02,E6,FD,B1,FA,91,FC
3 DATA E6,FC,D0,02,E6,FD,EA,E6,FA,D0,D6,E6,FB,A5,FB,C9,40,D0,CE,E6,FC,D0,02,E6,F
D,E6,FC,D0,02,E6,FD,A9,00,91,FC,60,00,EA,EA,E6,FA,D0,02,E6,FB,A0,00,EA,B1,FA
4 DATA 85,FC,E6,FA,D0,02,E6,FB,B1,FA,85,FD,EA,E6,FA,D0,02,E6,FB,B1,FA,D0,01,60,9
1,FC,EA,E6,FA,D0,02,E6,FB,A2,00,F0,D9,00
5 TRAP100
6 INPUT"(CLR)NOME";N$:INPUT"(CLR)[JARICA O [S]ALVA";L$:IFL$="C"THENBLOAD(N$).B0
:GOTO70
10 GRAPHIC0,1:INPUT"A,B":A,B:GRAPHIC1,1
12 FAST
15 POKE252,0:POKE253,69
20 FORG=OTO360:F=G*π/180
30 R=A*COS(B*COS(F))
40 LOCATE160,100:DRAW,ABS(R):G+90-180*(R<0)
42 Z=RDOT(0):W=RDOT(1)
44 IFF>0THENDRAW,X,YTOZ,W
46 X=Z:Y=W
50 NEXTG
55 BANK0:SYS4864
60 B=B+π/8:K=K+1:SCNCLR:IFK<20GOTO20
65 BSAVE(N$),B0,P17665TOP(PEEK(252)+256*PEEK(253)+1):SLOW
70 POKE250,0:POKE251,69:GRAPHIC1,1
80 FORK=1TO20:SCNCLR:BANK0:SYS4951:FORH=OTO100:NEXTH,K
90 GOTO70
100 RESUME46

```

Listato 5

Ad esempio introducendo a 80 e b = 2.75 il programma traccierà la famiglia di curve per il dato a e per b = 2.75, b = 3.14, b = 3.53 e così via.

Nelle figure 23, 24, e 25 sono indicate le prime tre curve di questa famiglia.

Per particolari famiglie di curve può convenire prendere per b un incremento diverso da $\pi/8 \approx 0.39$, prendendo ad esempio nella linea 60 $B = B + \pi/16$ oppure $B = B + 10$.

Programma Polare4

Il programma precedente ha l'incon-

```

1 FORA=4864T05001:READA$:POKEA,DEC(A$):NEXT
2 DATA EA,EA,E6,FC,D0,02,E6,FD,A9,00,85,FA,A9,20,85,FB,A0,00,EA,B1,FA,C9,00,F0,2
0,EA,A5,FA,91,FC,E6,FC,D0,02,E6,FD,A5,FB,91,FC,E6,FC,D0,02,E6,FD,B1,FA,91,FC
3 DATA E6,FC,D0,02,E6,FD,EA,E6,FA,D0,D6,E6,FB,A5,FB,C9,40,D0,CE,E6,FC,D0,02,E6,F
D,E6,FC,D0,02,E6,FD,A9,00,91,FC,60,00,EA,EA,E6,FA,D0,02,E6,FB,A0,00,EA,B1,FA
4 DATA 85,FC,E6,FA,D0,02,E6,FB,B1,FA,85,FD,EA,E6,FA,D0,02,E6,FB,B1,FA,D0,01,60,9
1,FC,EA,E6,FA,D0,02,E6,FB,A2,00,F0,D9,00
5 TRAP100
6 INPUT"(CLR)NOME";N$:INPUT"(CLR)[C]ARICA 0 [S]ALVA";L$:IFL$="C"THENBLOAD(N$),BO
:GOTO70
7 SCNCLR:LIST30:INPUT"(UP)(UP)R= ";E$:PRINT"(HOME)(DOWN)(DOWN)(DOWN)30 R="+E$:PR
INT"GOTO9(HOME)
8 POKE208,2:POKE842,13:POKE843,13:END
9 PRINT"(CLR)"TAB(18)"(UP)1,0.39(DOWN)":INPUT"(HOME)PASSO,INCREMENTO";P,I
10 GRAPHIC0,1:INPUT"A,B":A,B:GRAPHIC1,1
12 FAST
15 POKE252,0:POKE253,69
20 FORG=0T0360STEPP:F=G*PI/180
30 R=A*COS(B*COS(F))
40 LOCATE160,100:DRAW,ABS(R):G+90-180*(R<0)
42 Z=RDOT(0):W=RDOT(1)
44 IFF>0THENDRAW,X,YTOZ,W
46 X=Z:Y=W
50 NEXTG
55 BANK0:SYS4864
60 B=B+I:K=K+1:SCNCLR:IFK<20GOTO20
65 BSAVE(N$),BO,P17665TOP(PEEK(252)+256*PEEK(253)+1):SLOW
70 POKE250,0:POKE251,69:GRAPHIC1,1
80 FORK=1T020:SCNCLR:BANK0:SYS4951:FORH=0T0100:NEXTH,K
90 GOTO70
100 RESUME46

```

Listato 6

veniente di essere abbastanza lento, dato che per tracciare una sola curva della famiglia occorre quasi un minuto.

Se vogliamo accelerare il tracciamento, in modo da vedere rapidamente come varia la forma della curva al variare del parametro, cioè vogliamo realizzare un'animazione della famiglia, dobbiamo ricorrere al linguaggio macchina.

Con questo possiamo memorizzare le curve della famiglia man mano che vengono elaborate e alla fine visualizzare rapidamente queste memorizzazioni, in modo da ottenere l'animazione voluta.

Limitiamo l'esame della famiglia ad una ventina di curve e realizziamo la memorizzazione in forma compressa (cioè memorizzando solo i byte non nulli) sistemandola dopo il programma

Basic. Per ottenere ciò basta aggiungere al programma Polare3 alcune linee ottenendo così il programma descritto dal listato 4.

Se vogliamo vedere l'animazione delle Curve Butterfly procediamo come nel programma precedente.

Vedremo allora che, dopo il traccia-

re l'animazione su disco, in modo da poterla caricare e visualizzare in pochi secondi, quando lo si desidera.

Per ottenere ciò, e per accelerare l'elaborazione usando FAST, basta aggiungere al programma Polare4 le seguenti linee:

```

6 INPUT"(CLEAR)NOME";N$:INPUT"(CLEAR)[C]ARICA 0
[S]ALVA";L$:IFL$="C"THENBLOAD(N$),BO:GOTO70
12 FAST
65 BSAVE(N$),BO,P17665TOP(PEEK(252)+256*PEEK(253)+1):SLOW

```

Il programma così modificato è illustrato nel listato 5.

Ad esempio se vogliamo salvare l'animazione delle Curve Butterfly, lanciato il programma, introduciamo il nome BUTTERFLY e S.

L'elaborazione avviene come nel programma precedente, solo che viene effettuato anche il salvataggio su disco dell'animazione col nome Butterfly.

Se vogliamo invece visualizzare un'animazione precedentemente salvata su disco, lanciato il programma, basta introdurre il suo nome e C.

Programma Polare6

Nel programma precedente per modificare l'equazione della curva, il passo dell'anomalia e l'incremento del parametro, si devono riscrivere direttamente le linee 20, 30 e 60.

Se vogliamo prevedere la loro modifica da programma, è sufficiente aggiungere al programma Polare5 le linee seguenti:

```

7 SCNCLR:LIST30:INPUT"(UP)(UP)R= ";E$:PRINT"(HOME)
(DOWN)(DOWN)(DOWN)30 R="+E$:PRINT"GOTO9(HOME)
8 POKE208,2:POKE842,13:POKE843,13:END
9 PRINT"(CLEAR)"TAB(18)"(UP)1,0.39(DOWN)":
INPUT"(HOME)PASSO,INCREMENTO";P,I

```

mento lento delle curve della famiglia, successivamente esse vengono tracciate velocemente, realizzando così l'animazione. E questa animazione viene ripetuta continuamente finché fermiamo il programma premendo insieme RUN/STOP e RESTORE.

Programma Polare5

Dato che nel programma precedente l'elaborazione iniziale della famiglia richiede decine di minuti, conviene salva-

e sostituire le linee 20 e 60 con le seguenti:

```

20 FORG=0T0360STEPP:F=G*PI/180
60 B=B+I:K=K+1:SCNCLR:IFK<20GOTO20

```

Il programma così modificato è illustrato nel listato 6.

Ad esempio se vogliamo salvare l'animazione delle Curve Mosquito, lanciato il programma, introduciamo il nome MOSQUITO, S, l'equazione $A \cdot \sin(SIN(B \cdot F))$, il passo 2 e l'incremento 0.39.